SAND75-0116
Unlimited Release

A SIMPLIFIED CALIBRATION SEQUENCE FOR SINGLE-ATOM MASS SPECTROMETERS

J. A. Panitz

Prepared by Sandia Laboratories, Albuquerque, New Mexico 87115
and Livermore, California 94550 for the United States Energy Research
and Development Administration under Contract AT (29-1)-789

October 1975

Printed in the United States of America
Available from
National Technical Information Service
U. S. Department of Commerce
[illegible]
[illegible]
Price: Printed Copy [illegible]; Microfiche [illegible]

Sandia Laboratories

SF 2900 Q(7-73)

Issued by Sandia Laboratories, operated for the United States Energy Research and Development Administration by Sandia Corporation.

SAND75-0116
Unlimited Release
Printed March 1975

A SIMPLIFIED CALIBRATION SEQUENCE FOR SINGLE-ATOM MASS SPECTROMETERS

J. A. Panitz
Sandia Laboratories, Albuquerque, New Mexico 87115

ABSTRACT

A calibration procedure is described for single-atom, time-of-flight mass spectrometers which essentially applies a least-squares analysis to the measured parameters of voltage, distance and time in order to calculate the actual potential at the specimen surface, and any time delay associated with the measurements. A computer program numerically solves the required simultaneous equations by a variation of the Newton-Raphson method, quickly predicting the unknown desorption pulse amplitude and time delay to an accuracy of one part in 10^5.

The Field Desorption Spectrometer[1] and conventional Atom-Probe Field Ion Microscope[2] are time-of-flight analyzers which can determine the identity of field desorbed surface species if their energy, flight time and travel distance are known. Ion production is initiated by applying a short duration pulse of amplitude, P, to a suitable specimen pre-biased by a DC potential, V. If the duration of the pulse is made longer than the ion's travel time in the acceleration region near the specimen, the ion will quickly attain a kinetic energy determined only by the sum of the DC bias and applied pulse amplitude. That is:

$$\frac{1}{2} m v_t^2 = ne\ (V + P) \tag{1}$$

where ne is the ion's charge, and v_t its final, or terminal, velocity. Since the travel time of the ion in the acceleration region between the positively biased specimen and grounded cathode is always smaller than its drift time in field-free space, its travel time is, to a very good approximation, just:

$$\tau \simeq d/v_t \tag{2}$$

where d is the cathode-to-detector distance.

The ion selected for analysis is identified by its mass-to-charge ratio expressed as a function of the total voltage, travel time, and distance. Combining equations (2) and (3) gives the desired result:

$$\left(\frac{m}{n}\right)_o = K\ (V + P)\ \tau^2 \tag{3}$$

where

$$K \equiv \frac{0.193}{d^2}$$

and $\left(\frac{m}{n}\right)_o$ is expressed in atomic mass units (amu) if V_{dc} and V_{pulse} are measured in kilovolts, τ is measured in microseconds, and d is measured in meters.

However, because of the specimen's angstrom dimensions and the DC bias which is usually several kilovolts, the desorption pulse transmission line to the specimen cannot be terminated with its characteristic impedance at the specimen surface. This results in an actual pulse amplitude at the specimen surface of αP where α is called the "pulse factor," and represents the result of transmission line reflections. Experimental difficulties in determining the precise travel time to the needed accuracy of a few nanoseconds means that the actual travel time, τ differs from the measured time, t, by a "time-delay," δ. Rewriting equation (3) in terms of these new parameters gives:

$$\left(\frac{m}{n}\right)_o = K\ (V + \alpha P)(t \pm \delta)^2 \qquad (4)$$

Calibration of the Field Desorption Spectrometer or Atom-Probe Field Ion Microscope requires a determination of the pulse factor, α, and time delay, δ. The purpose of this report is to elaborate upon the details of one of two calibration methods reported previously.[3] Experience with both procedures has indicated that the one to be described is generally the more accurate and the simpliest to apply.

Assume that a species is produced whose identity is known either by field evaporating the substrate of a single-isotope specimen or by field desorbing a known adsorbate previously applied to the specimen surface. Further, assume that its charge state is known from field evaporation theory or comparison with a species observed previously so that its mass to charge ratio, M, can be calculated. Then M is just the quantity which equation (4) would predict if α and δ were known. One is interested, then, in minimizing the difference between M and $\left(\frac{m}{n}\right)_o$ for each species observed. In minimizing this difference for i species, the square of the difference must also be minimized. That is, the quantity

$$\sum_i \left[M - K\ (V + \alpha P)(t \pm \delta)^2 \right]^2 \tag{5}$$

is to be minimized with respect to the unknown quantities α and δ.* This can be accomplished by requiring that

$$\frac{\partial}{\partial\alpha} \sum_i \left[M - K\ (V + \alpha P)(t \pm \delta)^2 \right]^2 = 0 \tag{6}$$

and

$$\frac{\partial}{\partial\delta} \sum_i \left[M - K\ (V + \alpha P)(t \pm \delta)^2 \right]^2 = 0 \tag{7}$$

where V, P, and t may vary for each species, but are known. Performing the indicated differentiations gives

$$\sum_i \left[M - K\ (V + \alpha P)(t \pm \delta)^2 \right] P\ (t \pm \delta)^2 = 0 \tag{8}$$

and

*For convenience, the subscript i associated with M, V, P, and t has been omitted.

$$\sum_i \left[M - K\,(V + \alpha P)(t \pm \delta)^2 \right] (V + \alpha P)(t \pm \delta) = 0 \quad . \tag{9}$$

In order to calculate the best value of α and δ, equations (8) and (9) must be solved simultaneously. Their complexity requires a numerical solution, which may be obtained by direct application of the Newton-Raphson method. Reference 4 describes the general procedure in detail, which is applied here to the specific problem of determining α and δ. Equations (8) and (9) may be rewritten as follows:

$$K^2 \sum_i \left[(V + \alpha P)(t + \delta)^2 - \frac{M}{K} \right] \left[P\,(t + \delta)^2 \right] = 0 \tag{10}$$

$$K^2 \sum_i \left[(V + \alpha P)(t + \delta)^2 - \frac{M}{K} \right] \left[(V + \alpha P)(t + \delta) \right] = 0 \tag{11}$$

where V_i, P_i, and t_i are the DC voltage, pulse voltage, and observed travel time, respectively, which correspond to the i^{th} species observed. M is defined as the mass-to-charge ratio of the known calibration species, and the time delay δ may be negative.

Equations (10) and (11) are of the form:

$$\xi\,(\alpha, \delta) = 0 \tag{12}$$

$$\eta\,(\alpha, \delta) = 0 \quad . \tag{13}$$

Let

$$\alpha = \alpha_o + a \tag{14}$$

and

$$\delta = \delta_o + d \tag{15}$$

where α_o and δ_o are approximate values for the pulse factor and time delay, and a and d are the appropriate correction terms. Substitution of equations (14) and (15) into equation (12) and (13) and expanding in a Taylor series about α and δ, gives

$$\xi\,(\alpha_o + a\,,\,\delta_o + d) \simeq \xi\,(\alpha_o\,,\,\delta_o) + a\left(\frac{\partial\xi}{\partial\alpha}\right)_o + d\left(\frac{\partial\xi}{\partial\delta}\right)_o = 0 \qquad (16)$$

$$\eta\,(\alpha_o + a\,,\,\delta_o + d) \simeq \eta\,(\alpha_o\,,\,\delta_o) + a\left(\frac{\partial\eta}{\partial\alpha}\right)_o + d\left(\frac{\partial\eta}{\partial\delta}\right)_o = 0 \qquad (17)$$

where terms in higher powers of a and d as well as their products have been neglected, since the correction terms, a and d are assumed small. The subscript "o" attached to each derivative corresponds to the subscript of α and δ, and indicates that the derivative is to be evaluated for these values of α and δ.

Equations (16) and (17) may be solved by the method of determinants for the initial correction terms a and d. These correction terms to α_o and δ_o are just

$$a = \frac{\begin{vmatrix} -\,\xi(\alpha_o\,,\,\delta_o) & \left(\frac{\partial\xi}{\partial\delta}\right)_o \\ -\,\eta(\alpha_o\,,\,\xi_o) & \left(\frac{\partial\eta}{\partial\delta}\right)_o \end{vmatrix}}{\begin{vmatrix} \left(\frac{\partial\xi}{\partial\alpha}\right)_o & \left(\frac{\partial\xi}{\partial\delta}\right)_o \\ \left(\frac{\partial\eta}{\partial\alpha}\right)_o & \left(\frac{\partial\eta}{\partial\delta}\right)_o \end{vmatrix}} \qquad (18)$$

$$d = \frac{\begin{vmatrix} \left(\frac{\partial \xi}{\partial \alpha}\right)_{o} & -\xi(\alpha_{o}, \delta_{o}) \\ \left(\frac{\partial \eta}{\partial \alpha}\right)_{o} & -\eta(\alpha_{o}, \delta_{o}) \end{vmatrix}}{\begin{vmatrix} \left(\frac{\partial \xi}{\partial \alpha}\right)_{o} & \frac{\partial \xi}{\partial \delta}_{o} \\ \left(\frac{\partial \eta}{\partial \alpha}\right)_{o} & \frac{\partial \eta}{\partial \delta}_{o} \end{vmatrix}} \quad . \tag{19}$$

A corrected value for α and δ can now be obtained from equations (14) and (15), i.e.,

$$\alpha_1 = \alpha_o + a \tag{20}$$

$$\delta_1 = \delta_o + d \quad . \tag{21}$$

These new values can be used to obtain a further correction, and the iteration process repeated as often as necessary to obtain the required accuracy. The calculation is terminated when the correction terms, a and d, are smaller than some assumed value. The simultaneous solution of equations (10) and (11) by this method requires the use of the following quantities:

$$\xi(\alpha_1 \ \delta) \equiv K^2 \sum_i \left[(V_i + \alpha P_i)(t_i + \delta)^2 - \frac{M_i}{K}\right]\left[(V_i + P_i)(t_i + \delta)\right] = 0 \tag{22}$$

$$\frac{\partial \xi}{\partial \alpha} = 2K^2 \sum_i \left[(V_i + \alpha P_i)(t_i + \delta)^2 - \frac{M_i}{2K}\right]\left[P_i \ (t_i + \delta)\right] \tag{23}$$

$$\frac{\partial \xi}{\partial \delta} = 3K^2 \ \Sigma\left[(V_i + \alpha P_i)(t_i + \delta)^2 - \frac{M_i}{3K}\right]\left[V_i + \alpha P_i\right] \tag{24}$$

$$\eta(\alpha,\delta) \equiv K^2 \sum_i \left[(V_i + \alpha P_i)(t_i + \delta)^2 - \frac{M_i}{K}\right]\left[P_i \; (t_i + \delta)^2\right] = 0 \tag{25}$$

$$\frac{\partial\eta}{\partial\alpha} = K^2 \sum_i \left[P_i \; (t_i + \delta)^2\right]^2 \tag{26}$$

$$\frac{\partial\eta}{\partial\delta} = (2K)^2 \sum_i \left[(V_i + P_i)(t_i + \delta)^2 - \frac{M_i}{2K}\right]\left[P_i \; (t_i + \delta)\right] . \tag{27}$$

The computer program which follows calculates the correction terms a and d that correspond to the given initial values α_o and δ_o. The resulting corrected values for α and δ are then used to generate a second-order correction and the iteration process repeated until α and δ, at succeeding cycles, differ by less than 1.0×10^{-5}. In practice, less than twenty cycles are required to achieve this accuracy. The resulting numbers are then rounded to two decimal places to give final values for the pulse factor and the time delay. One note of caution is in order. The effective pulse amplitude, αP, may actually be different for the different observed species since a species of large m/n will remain in the acceleration region near the specimen for a longer time than a species of small m/n. Thus, its kinetic energy will be affected more by the exact shape of the desorption pulse. This means that ideally, one should calibrate using known species having mass-to-charge ratios close to those of the unknown species of interest, otherwise an average value for α and δ is all that one can hope to achieve.

```
      PROGRAM CALIB(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C
C ITERATION SCHEME FOR DETERMINING THE PULSE CORRECTION AND TIME DELAY
C FOR THE FIELD DESORPTION MASS SPECTROMETER.
C
      DIMENSION T(500),V(500),P(500),CM(500)
    1 FORMAT(3F10.4,I3)
    2 FORMAT(4F10.4)
    8 FORMAT(' ',22X,I4,18X,F6.3,18X,F8.5,18X,F10.8)
   11 FORMAT('0',40X,'M/N = ',F7.4,'(V + (',F5.4,')P)(T + (',F7.5,'))**2
     1'///29X,70('*')///)
   12 FORMAT(1H1,32X,'PARAMETERS FOR FDS CALIBRATION USING THE ASSIGNED
     1CALIBRATION SPECIES'///25X,'PULSE FACTOR = ',F6.4,05X,'TIME DELAY
     2= ',F8.4,' NANOSECONDS',04X,'DISTANCE = ',F6.4,' METERS'////04X,
     3'TIP VOLTAGE',05X,'PULSE-CABLE VOLTAGE',05X,'PULSE AMPLITUDE',07X,
     4'MEAS TIME',    11X,'CALC TIME',     08X,'CALC TRAVEL TIME',05X,
     5'CALC M/N'/06X,'(KVDC)',14X,'(KVDC)',15X,'CALC (KV)',09X,'(NANOSEC
     6ONDS)',06X,'(NANOSECONDS)',06X,'(NANOSECONDS)',08X,'(AMU)'//)
   13 FORMAT(I2)
   17 FORMAT('1'// 09X,'ITERATION SCHEME FOR DETERMINING THE PULSE CORRE
     1CTION AND THE TIME CORRECTION FOR THE FIELD DESORPTION SPECTROMETE
     2R'/31X,'INITIAL PULSE FACTOR = ',F4.2,05X,'INITIAL TIME DELAY = ',
     3F6.3,' MICROSECONDS'///18X,'ITERATION CYCLE',05X,'CORRECTED PULSE
     4FACTOR',05X,'CORRECTED TIME DELAY',06X,'MEAN-SQUARE DEVIATION'//)
   19 FORMAT(06X,F6.3,14X,F5.3,18X,F5.3,14X,F6.1,13X,F6.1,14X,F6.1,11X,
     1F6.2)
   21 FORMAT(01X///28X,70('*')//54X,'END OF CALIBRATION')
C
      READ(5,1) A,D,DI,KL
C A IS THE PULSE FACTOR, D IS THE TIME DELAY (NANOSECONDS), DI IS THE
C DISTANCE (METERS),AND KL IS THE NUMBER OF DATA CARDS TO BE READ BY
C FORMAT 2.
C
      AINIT = A
      DINIT = D
      DO 5 I=1,KL
C
      READ(5,2) CM(I),P(I),V(I),T(I)
C CM(I) IS THE CALIBRATION SPECIES ASSIGNED TO EACH MEASURED TRAVEL TIME,
C P(I) IS THE PULSE-FORMING CABLE VOLTAGE (KV) CORRESPONDING TO A TIP
C VOLTAGE OF V(I) IN KV, AND THE MEASURED TRAVEL TIME, T(I), IN NANOSECONDS.
C
C  NOTE- THE PULSE FORMING CABLE VOLTAGE IS TWICE THE IDEAL PULSE AMPLITUDE.
C
    5 T(I)=T(I)*1.0E-03
      A = AINIT
      D = DINIT
      D=D*1.0E-03
      CON=(2.*.160206)/(1.65979*(DI**2))
      WRITE(6,17) A,D
      KOUNT=0
    4 KOUNT=KOUNT+1
      PHI=0.0
      PHIA=0.0
```

```
      PHID=0.0
      PSI=0.0
      PSID=0.0
      SSQ=0.0
      DO 6 I=1,KL
C
C   CALIBRATION EQUATIONS
      C=CM(I)/CON
      TD=T(I)+D
      AP=A*P(I)
      AD=(V(I)+AP)*(TD**2)
      W=(AD-C)*P(I)*(TD**2)
      W1=(P(I)*(TD**2))**2
      W2=(AD-(C/2.))*P(I)*TD
      X=(AD-C)*(V(I)+AP)*TD
      X2=(AD-(C/3.))*(V(I)+AP)
      PHI=PHI+W
      PHIA=PHIA+W1
      PHID=PHID+W2
      PSI=PSI+X
      PSID=PSID+X2
C
C CALCULATE THE SUM OF THE SQUARES OF THE MASS DIFFERENCES
      SSQ=SSQ+((CM(I)-(CON*AD))**2)
C
    6 CONTINUE
      SSQ=SSQ/KL
      PHI=PHI*CON**2
      PHIA=PHIA*CON**2
      PSIA=PHID*2.*CON**2
      PHID=PHID*4.*CON**2
      PSI=PSI*CON**2
      PSID=PSID*3.*CON**2
      Q=PHIA*PSID-PHID*PSIA
C
C CALCULATE CORRECTION TO PULSE FACTOR
      CORA=(PHID*PSI-PHI*PSID)/Q
C
C
C CALCULATE CORRECTION TO TIME DELAY
      CORD=(PHI*PSIA-PHIA*PSI)/Q
C
      A1=A
      D1=D
      WRITE(6,8) KOUNT,A,D,SSQ
      A=A+CORA
      D=D+CORD
      E=ABS(A-A1)
      F=ABS(D-D1)
C
C TEST FOR NUMBER OF ITERATION CYCLES
      IF(KOUNT.GE.200) GO TO 10
C
C
```

```
C TEST FOR CONVERGENCE OF ITERATION SCHEME
      IF(E.LE..00001.AND.F.LE..00001) GO TO 10
C
      GO TO 4
   10 WRITE(6,11) CON,A,D
      D=D*1.0E+03
      WRITE(6,12) A,D,DI
      DO 20 I=1,KL
      P1=A*P(I)
      V1=(V(I)+P1)*CON
      AM=V1*((T(I)+(D*1.0E-03))**2)
      TM=T(I)*1.0E+03
      TC=CM(I)/V1
      TA=SQRT(TC)*1.0E+03
      TC=TA-D
   20 WRITE(6,19) V(I),P(I),P1,TM,TC,TA,AM
      WRITE(6,21)
      STOP
      END

       0.5         0.0         .1180  13
     1.000       2.000       1.521        170.0
     1.000       2.000       2.000        157.0
     1.000       2.000       2.503        146.0
     1.000       2.000       3.001        137.0
     1.000       2.000       3.505        129.5
     1.000       2.000       4.002        124.0
     1.000       2.000       4.503        119.0
     1.000       2.000       5.002        112.0
     1.000       2.000       5.508        108.5
     1.000       2.000       6.012        105.0
     1.000       2.000       6.505        103.5
     1.000       2.000       7.016         99.0
     1.000       2.000       8.004        093.0
```

APPENDIX I

1. Computer Program Legend

A - The pulse factor, α. For an ideally terminated cable, A = 0.5 exactly one-half of the pulse-forming cable voltage, P.

D - The time delay, δ, in nanoseconds. The time delay can be positive or negative and is the time between the time from which all measurements of travel time are made (fiducial), to the time the desorption event actually occurs.

DI - The drift distance in meters between cathode and detector; a constant. If DI is also to be determined, an initial best estimate of DI is used to predict α and δ for one known species then incremented until, with the corresponding values of α and δ, the correct identity of a second known species is also predicted.

KL - The number of separate events used for the calibration sequence, which equals the number of date cards to be read by format 2.

CM(I) - The ith calibration species, M_i, in amu.

P(I) - The corresponding pulse-forming cable voltage, in kilovolts.

V(I) - The corresponding DC specimen bias, in kilovolts.

T(I) - The corresponding measured travel time in nanoseconds.

AINIT - The initial value chosen for the pulse factor, usually the ideal value, 0.5.

DINIT - The initial value chosen for the time delay, usually the ideal value, 0.0 nanoseconds.

CON - The constant, K, of equation (3).

KOUNT - The number of the iteration cycle being performed.

C - The constant M_i/K in equations (22)-(27).

PHI - Equation (25).

PHIA - Equation (26).

PHID - Equation (27).

PSI - Equation (22).

PSID - Equation (24).

PSIA - Equation (23).

SSQ - The mean of the sum-of-the-squares of the differences between actual and calculated mass-to-charge ratios in each iteration cycle.

CORA - The correction to the Pulse factor, Equation (18).

CORD - The correction to the time delay, Equation (19).

E - The absolute value of the difference between the pulse factors of two successive iteration cycles.

F - The absolute value of the difference between the time delays of two successive iteration cycles.

2. Program Output

The following is the output of the calibration program for the input data shown after the main program (above). Note that the "calculated travel time" is the actual travel time of the species, whereas the "calculated time" includes the time delay, and is equal to the input data shown under "measured time." "Pulse amplitude-calculated" includes the pulse factor, and "calculated m/n" is the calculated value of the input species using the calibration parameters. The input species was, again, m/n = 1.0 (H^+).

ITERATION SCHEME FOR DETERMINING THE PULSE CORRECTION AND THE TIME CORRECTION FOR THE FIELD DESORPTION SPECTROMETER

INITIAL PULSE FACTOR = .50 INITIAL TIME DELAY = 0.000 MICROSECONDS

ITERATION CYCLE	CORRECTED PULSE FACTOR	CORRECTED TIME DELAY	MEAN-SQUARE DEVIATION
1	.530	0.00000	.00424352
2	.575	-.00503	.00019736
3	.555	-.00467	.00017727
4	.556	-.00468	.00017727

M/N = 13.864((V + (.5555)P)(T + (-.00468))**2

****

PARAMETERS FOR FDS CALIBRATION USING THE ASSIGNED CALIBRATION SPECIES

PULSE FACTOR = .5555 TIME DELAY = -4.6783 NANOSECONDS DISTANCE = .1180 METERS

TIP VOLTAGE (KVDC)	PULSE-CABLE VOLTAGE (KVDC)	PULSE AMPLITUDE CALC (KV)	MEAS TIME (NANOSECONDS)	CALC TIME (NANOSECONDS)	CALC TRAVEL TIME (NANOSECONDS)	CALC M/N (AMU)
1.521	2.000	1.111	170.0	170.2	165.5	1.00
2.009	2.000	1.111	157.0	156.9	152.3	1.00
2.503	2.000	1.111	146.0	146.0	141.3	1.00
3.001	2.000	1.111	137.0	137.1	132.4	1.00
3.505	2.000	1.111	129.5	129.7	125.0	1.00
4.002	2.000	1.111	124.0	123.5	118.8	1.01
4.503	2.000	1.111	119.0	118.0	113.3	1.02
5.002	2.000	1.111	112.0	113.3	108.6	.98
5.508	2.000	1.111	108.5	109.1	104.4	.99
6.012	2.000	1.111	105.0	105.3	100.6	.99
6.505	2.000	1.111	103.5	102.0	97.3	1.03
7.016	2.000	1.111	99.0	98.9	94.2	1.00
8.004	2.000	1.111	93.0	93.6	89.0	.99

****

END OF CALIBRATION

REFERENCES

1. J. A. Panitz, J. Vac. Sci. Technol. 11, 206 (1974).

2. E. W. Müller, J. A. Panitz and S. B. McLane, Rev. Sci. Instr. 39, 83 (1968).

3. J. A. Panitz, S. B. McLane, and E. W. Müller, Rev. Sci. Instr. 40, 1321 (1969).

4. J. B. Scarborough, Numerical Mathematical Analysis (Johns Hopkins Press, Baltimore, 1966), p. 215.

Distribution:
5000 A. Narath
Attn: 5200 E. H. Beckner
5600 A. Y. Pope
5700 J. H Scott
5800 R. S. Claassen
5100 J. K. Galt
5110 F L. Vook
5114 J. E. Houston
5114 J. A. Panitz (24)
3141 L. S. Ostrander (5)
3151 W. L. Garner (3)
For: ERDA/TIC (Unlimited Release)
ERDA/TIC (25)
(3171-1 R. P. Campbell)
8266 E. A. Aas (2)